从物种增长的 Malthus 模型到混沌

5121209117 徐小博

5122409009 王力功

一、　实验目的

本实验涉及函数的迭代、不动点和有关的作图。实验通过对一个简单的一维二次函数映射，即 Logistic 映射的讨论，介绍了用数值迭代、蛛网迭代和密度分布等方法来研究混沌；说明了混沌的倍周期分叉、遍历性和某些普适结构；进而说明计算机和数学的结合在科学研究中的重要性。

二、　问题的提出

“混沌”这个名词出现在生活的各个领域，不仅出现在数学、物理和生物等自然科学中，而且出现在金融、经济和管理等社会科学中；甚至出现在文学和艺术的范畴：从宇宙的形成到龙卷风的产生、从东南亚金融危机爆发到“侏罗纪公园”中的恐龙出现。“混沌”很难用一两句话描述清楚，但是往往可以通过一些并不复杂的例子进行考察，一个著名的例子是 Logistic 映射。

Logistic 映射源于生物数学中的物种增长模型。物种的生长与衰亡是自然界最基本的现象，对物种群体数量的研究已经有相当长久的历史。我们将从简单的离散数学模型开始，进而讨论由此引起的复杂而有趣的现象。

三、　问题的讨论和分析

1. 对 Logistic 映射$f(x)=\alpha x(1-x)$，取α依次属于区间$(c_k,\ c_{k+1}), k=1,2,3,4$。然后取值在 3.6 附近和接近 4；任给一个初始值x_0，用数值迭代方法求序列$\{x_n\}$来考察其趋向；进而在周期 3 窗口取α值为 3.83 到 3.84 之间，考察由x_0出发所得$\{x_n\}$的趋向，再通过适当增加α的值，得到分叉到周期 6 的情况；能否再得到分叉到周期 12 的情况？对于上述结果可以在$n-x_n$平面上作图考考察，不过注意 n 应该取得足够大，才能有足够多的点（数百个或更多）看出变化趋势。

【问题的解答】

由书中所述，我们先在题目给定的四个区间内分别取了四个值，验证了确实是二分叉、四分叉、八分叉、十六分叉。取定初始值$x_0=0.314$，用数值迭代方法，我们完成了

Mathematica 程序，如下所示：

```
f[{x_, α_}] := {α * x (1 - x), α};
g[x0_, α_, n_] := (Transpose@NestList[f, {x0, α}, n])[[1]];

αs = {3.2, 3.5, 3.55, 3.566, 3.61, c1, c2, c3, 3.99};
x0 = 0.314;
Row@Table[ListPlot[g[x0, α, 1000], AxesOrigin → {0, 0},
   AspectRatio → 4, ImageSize → 110,
   PlotRange → {Automatic, {0.0, 1.0}}, TicksStyle → 14,
   AxesLabel → {"n" ~ Style ~ 17, "xn" ~ Style ~ 20},
   Ticks → {Range[0, 1000, 500], Range[0, 1, 0.1]},
   PlotLabel → ("α=" <> ToString@α) ~ Style ~ 18], {α, αs}]
```

然后，我们在$n-x_n$平面上做出了以上四个区间所对应的四个值（区间中任意选定的值）的趋向，如图一的前四个平面所示：



图一

进一步的，我们取 3.6 附近的值 3.61，和 4 附近的值 3.99。我们发现，在 α = 3.61 时已经发生了混沌现象，这说明 3.61 已经超过了混沌的临界点c_∞。

与此同时，我们注意到，α = 3.61 时混沌现象只覆盖了区间的一部分，而 α = 3.99 时混沌现象却覆盖了整个区间。

为了研究c_∞右侧的现象：周期 3、周期 6、周期 12 现象，我们在周期 3 窗口取α 值为 3.83 到 3.84 之间，取了 3.835 这个值，然后作图。从图一中可知，当$n\to\infty$时，数列$\{x_n\}$趋向于 3 个不同的极限值，从而验证了周期 3 的现象。

为了得到分叉到周期 6 的情况，我们通过二分法，适当增加α的值，找到了周期 3 过渡

到周期 6 的分界点。该二分法的过程如下图所示，这三组图代表了原来三分叉的每一个分叉是如何加倍的：



从图中我们看到，在 α = 3.8416 之前，当 n 足够大时，$\{x_n\}$都有收敛的趋势。而 α = 3.8416 的情况下，$\{x_n\}$趋向于了六个不同的极限值。同理，我们找到了周期 12 的分界点，为3.8477。

在图中，我们也可以看到，我们取了 n = 1000，这是一个相当大的数字，方便我们清晰地观察出收敛情况。

2. 对任务 1 中的初值x_0作一点增加，例如增加 0.0001 或更小，然后对不同的 α 取值，比较x_{100}，x_{500}，x_{1000}和更后的一些项与对应的原来这些序号的项的差距。这情况说明什么？

【问题的解答】

我们根据和题 1 中的初值增加了 0.0001，对比了不同 α 的情况下，数列的第 n 个值和原来的差值，n 分别取了 100 至 4000。

Mathematica 代码如下所示：

```
ns = {100, 500, 1000, 2000, 3000, 4000};
Grid[
 ({{"α\\n", Grid[{ns}, Spacings → {3.5, 0}]}} ~ Join ~
   Table[
    {α,
     Grid@{AccountingForm@PaddedForm[#, {5, 5}] & /@
        (g[x0 + 0.0001, α, Max@ns][[ns]] -
          g[x0, α, Max@ns][[ns]])}}, {α, αs}]),
 Frame → All]
```

结果如下表所示：

α\n	100	500	1000	2000	3000	4000
3.2	0.00000	0.00000	0.00000	0.00000	0.00000	0.00000
3.5	0.00000	0.00000	0.00000	0.00000	0.00000	0.00000
3.55	0.00413	0.00000	-0.00000	-0.00000	-0.00000	-0.00000
3.566	-0.00100	-0.00000	0.00000	0.00000	0.00000	0.00000
3.61	-0.04968	-0.04237	-0.02412	-0.00303	-0.08914	-0.01446
3.835	0.34244	0.46412	0.34244	0.46412	-0.80656	0.34244
3.8416	0.46927	-0.80973	0.47110	-0.80977	0.33607	0.47110
3.8477	0.00047	-0.00059	0.00041	-0.00059	0.00386	0.00041
3.99	-0.44320	-0.69757	0.22716	-0.32528	0.58300	-0.03058

从表中可以看出，在c_∞之前（混沌出现之前），也就是表中的前四个 α 值，初始值的细微变化对之后的值的影响微乎其微。

而在c_∞之后，产生了混沌现象。初始值的细微变化将剧烈影响数列的演化趋势（即表中显示差值的绝对值远远大于零），从而验证了混沌效应。

3. 对任务 1 中的 α 取值，用蛛网迭代的方法，自己编程进行计算机作图，考察由x_0出发的轨道情况。

【问题的解答】

我们写了 Mathematica 程序，完成了任务 1 中所有 α 值对应的蛛网迭代的作图。程序代码如下：

```
h[{{{x_, x_}, {x_, y_}}, α_}] :=
  {{{y, y}, {y, f[{y, α}][[1]]}}, α};
p[x0_, α_, n_] :=
  (Transpose@NestList[h, {{{x0, x0}, {x0, f[{x0, α}][[1]]}},
        α}, n])[[1]]~Flatten~1;

Row@Table[Show[
    ListLinePlot[p[x0, α, 10000][[9900 ;;]], AspectRatio → 1,
     Frame → False, ImageSize → 250, AxesOrigin → {0, 0},
     PlotLabel → "α=" <> ToString@α,
     PlotRange → {{0, 1}, {0, 1}},
     PlotStyle → Directive[Thickness@0.01, Black]],
    Plot[x, {x, 0, 1}, PlotStyle → Blue],
    Plot[f[{x, α}][[1]], {x, 0, 1}, PlotStyle → Blue]
  ], {α, {2.5}~Join~αs}]
```

蛛网作图效果如下：





为了方便观察，减小干扰，我们在作图时去掉了最初的 100 个点。

可以从图中看出，轨道的周期性情况，即最终的极限轨道和 $y = x$ 直线的交点数等于极限点数。

在 α = 2.5 时，由于没有出现分叉情况，所以数列收敛到抛物线与直线的交点。

在 α = 3.2、3.5、3.55、3.566 时，交点数量分别为 2、4、8、16 个。从而验证了倍周期现象（由于极限点距离太近，所以图中黑线有所重叠）。

同理，在 α = 3.835、3.8416、3.8477 时，图中交点数验证了周期 3、周期 6、周期 12 现象。

在 α = 3.61、3.99 时，分别产生了覆盖部分区间和覆盖所有区间的混沌现象。

4. 对任务 1 中的 α 取值，用密度图的方法，自己编程进行计算机作图（N = 10000），考察由x_0出发的$\{x_n\}$的分布。

【问题的解答】

我们完成了 Mathematica 程序：

```
Row@Table[Histogram[g[x0, α, 10000], 20, "Probability",
    PlotLabel → "α=" <> ToString@α, ImageSize → 350,
    AspectRatio → 0.3,
    AxesLabel → {"x"~Style~15, "Pr"~Style~15},
    PlotRange → {{0, 1}, {0, 0.6}}, AxesOrigin → {0, 0},
    Frame → True], {α, αs}]
```

当我们取 N = 10000 时，可以看出：当数列有周期性情况存在时，概率分布会集中在某几个有限的点上。从有限点的个数和对应的概率可以看出目前是周期几的分叉。具体作图如下所示：



从图中可以看出，未产生混沌现象的 α 值所对应的图中，只有有限个点存在概率分布，由此也印证了周期性的存在。

而在产生混沌现象的 α 值所对应的图中，存在某个或多个区间，均有概率分布。这体现了混沌现象的遍历性。

5. 编出计算程序，在 $\alpha - x$ 平面上画出 Logistic 模型的分叉图（其中 x 是稳定的周期点）

【问题的解答】

我们取定 α 遍历的适当步长 0.003，然后再取数列的后若干项（数列迭代 1000 次的后 300 项），写出了程序：

```
maxn = 1000; lastn = 300; minα = 2.9; maxα = 4; stepα = 0.003;
ListPlot[
 Table[{α, #} & /@ g[x0, α, maxn][[maxn - lastn ;;]],
   {α, Range[minα, maxα, stepα]}] ~ Flatten ~ 1,
 ImageSize → 1000, AspectRatio → 0.5,
 AxesLabel → {"α" ~ Style ~ 20, "x" ~ Style ~ 20},
 Ticks → {Range[minα, maxα, 0.05], Range[0, 1, 0.1]},
 AxesOrigin → {minα, 0},
 PlotLabel → "Logistic模型分叉图" ~ Style ~ 20,
 PlotStyle → PointSize@0.0015,
 PlotRange → {{minα, maxα}, {0, 1}}]
```

再作图，画出了 $\alpha - x$ 平面上 logistic 模型的分叉图，如下所示：



加细步长至 0.001，取后 800 项，图中点会变得非常稠密，如下图所示。我们会看到非常神奇而优美的混沌现象。



令我们非常好奇的是，在图中有一条线（在上图中用红色标出），我们无法解释他是什么。不管在什么步长下，这条线都非常的清晰。

6. 对映射

$$f(x) = \lambda \sin(\pi x), \ \lambda \in (0,1]$$

试考察当 λ 逐渐增大时，有没有倍周期分叉的情况出现？求出第一个分叉值和第二个分

叉值，利用 Feigenbaum 常数估计第三个分叉值和混沌可能在何时出现，验证第三个分叉值。

7. 试在 $\lambda - x$ 平面上画出该映射的分叉图，将它与 Logistic 映射的分叉图比较。

【对问题 6 & 7 的解答】

按照我们在前几个任务中的相同做法，我们写出了 Mathematica 程序：

```
f[{x_, λ_}] := {λ*Sin[Pi*x], λ};
g[x0_, λ_, n_] := (Transpose@NestList[f, {x0, λ}, n])[[1]];

maxn = 1000; lastn = 800; minλ = 0.7; maxλ = 1; stepλ = (maxλ - minλ) / 500;
ListPlot[
 Table[{λ, #} & /@ g[x0, λ, maxn][[maxn - lastn ;;]],
   {λ, Range[minλ, maxλ, stepλ]}] ~ Flatten ~ 1, ImageSize → 1000,
 AspectRatio → 0.5, AxesLabel → {"λ" ~ Style ~ 20, "x" ~ Style ~ 20},
 Ticks → {Range[minλ, maxλ, (maxλ - minλ) / 20], Range[-1, 1, 0.1]},
 AxesOrigin → {minλ, 0}, PlotLabel → "λ映射分叉图" ~ Style ~ 20,
 PlotStyle → PointSize@0.0015, PlotRange → {{minλ, maxλ}, {0, 1}},
 GridLines → {Range[minλ, maxλ, (maxλ - minλ) / 20], Range[-1, 1, 0.1]},
 GridLinesStyle → Dashed]
```

我们做出了 $\lambda - x$ 平面迭代式的分叉图：



由于从 α > 0.7 左右开始，才是我们真正关心的部分，所以我们将这一部分在图中放大进行观察：



从图中我们可以看出，λ 从 0 增加到 1 的过程中，有倍周期分叉的情况出现。我们可以求出，第一个分叉值为 0.7192，第二个分叉值为 0.8327。

利用 Feigenbaum 常数，我们估计出第三个分叉值为

$$C_3 = C_2 + \frac{C_2 - C_1}{\delta} \approx 0.857008$$

而在图中根据二分法实际找出的第三个分叉值为 0.8585，如下图所示：



由此可见，预测值和实际值非常接近。

由于分叉值数列$\{C_k\}$存在极限，根据 Feigenbaum 常数，我们求得了这个极限值为

$$C_\infty = C_1 + (C_2 - C_1) \cdot \frac{\delta}{\delta - 1} = 0.863633$$

但这不是最终的实际值。如何求出混沌现象发生的实际值呢？我们依然使用了二分法，做出了$n - x_n$平面图。根据这个方法，我们找到了近似C_∞的值，在 0.865 附近。可以看出，计算出的值和实际值也是相当接近的。

8. 分工情况

王力功：分析问题，数学建模，编程实现及作图

徐小博：组织文章结构，整理数据及编写，结果分析